
Einleitung

· Zurück zum Anfang
Das Arbeiten mit Word-Tabellen in einem VBA-Programm ist eigentlich eine erfreuliche Sache. Solange die Tabellen nicht ineinander verschachtelt sind, was erst ab Word2000 möglich ist, ist die Handhabung recht einfach, konsistent und einleuchtend. Problematischer wird es, wenn die Tabelle Zellen enthält, welche unter einander verbunden sind oder wenn eine Tabelle vorliegt, welche frei gezeichnet wurde. Viele VBA-Funktionen, wie z. B. das Auslesen der Zellhöhe, stehen dann nämlich nicht mehr bzw. nur noch bedingt zur Verfügung.

Erfahrungsgemäss haben Tabellen einen hohen Stellenwert beim Benutzer, das will heissen man greift gerne auf Tabellen zurück. Dieses Skript soll an Hand von konkreten Codebeispielen aufzeigen, wie man die häufigsten Problemstellungen im Zusammenhang mit Word-Tabellen angehen kann.

· Eine Einschränkung muss trotz allem schon Eingangs gemacht werden Bei grösseren und/oder komplexeren Tabellen ist es äusserst ineffizient zuerst eine Tabelle zu erstellen und anschliessend diese Tabelle mit Daten zu füllen. Mit unter können bei diesem Ansatz mehrere Minuten vergehen, bis Ihr Makro am Ziel angelangt ist. Um Dimensionen schneller ist es, zuerst die Daten zusammenzusetzen. Verwenden Sie dazu am besten eine String-Variable. Dieser String enthält typischerweise Tabulatoren, um die Spalten und Absatzmarken um die Zeilen anzudeuten. Schreiben Sie dann diesen String in das Word-Dokument und verwenden Sie schliesslich die mächtige ConvertToTable-Methode, um diese rudimentär vorformatierten Daten in eine anschauliche und echte MS Word Tabelle zu transferieren. Wie Sie dies konkret erreichen, wird im Kapitel Eine Tabelle mit VBA erstellen, ausführlich erklärt.

Das Objekt-Modell von Tabellen in MS Word

· Zurück zum Anfang
Das Objekt-Modell von Tabellen birgt keine grossen Überraschungen; es ist ziemlich einleuchtend aufgebaut. Hier einige wichtige Grundsätze des Table Objects in MS Word:

· Jeder Dokumentteil (Hauptteil, Kopf-/Fusszeile, Textfeld) eines Word-Dokuments kann 1 - n Tabellen beinhalten. Jede dieser Tabellen kann über den Index angesprochen werden. Dieser Index ist sequentiell und aufsteigend. Einen aussagekräftigen Namen besitzt eine Tabelle nicht und man kann auch keinen vergeben, es sei denn indirekt über eine Textmarke, welche man an beliebiger Stelle in einer Tabelle platziert.

· Eine Tabelle besteht aus 1 - 63 Spalten, welche Unterobjekte der Tabelle darstellen. Diese Objekte werden in VBA als Columns angesprochen. Die Zählung ist strikt numerisch und beginnt mit 1.

· Im Weiteren besteht eine Tabelle aus 1 - n Zeilen, wobei n fast beliebig gross sein kann. Auch diese Objekte - in Word-VBA Rows genannt - sind Elemente der Tabelle. Die Zählung ist strickt numerisch und beginnt mit 1.

· Die kleinsten Elemente der Tabelle schliesslich, sind die Zellen. Diese werden in VBA als Cells angesprochen. VBA erweist sich als flexibel, können Zellen doch sowohl als Elemente von Zeilen, Spalten oder sogar als Elemente von Tabellen direkt angesprochen werden. Die Syntax ist allerdings unterschiedlich. Spricht man eine Zelle über eine Zeile oder eine Spalte an, dann hat die Zelle eine Sequenznummer; werden Zellen dagegen direkt als Elemente von Tabellen angesprochen, dann muss man deren Koordinaten angeben. Die Zählung ist strickt numerisch. Die erste Zelle ist die Zelle, welche oben links in der Tabelle thront, dann geht die Zählung in Leserichtung weiter, also von links nach rechts und dann von oben nach unten.

· Tables, Rows, und Cells verfügen über ein Range-Objekt. Damit ist der Zugriff auf diese Objekte einfach und effizient. Das Columns-Objekt, dagegen verfügt aber über kein Range, da die Daten in einer Spalte nicht der Leserichtung folgen. Damit stellt eine Spalte kein zusammenhängender Bereich dar, müsste man doch den Pinsel beim bemalen einer Spalte (immer noch in Leserichtung betrachtet) immer wieder neu ansetzen bzw. absetzen.

· Bei der Rückgabe des Cell-Ranges ist zu beachten, dass der Text in diesem Range immer mit zwei Steuerzeichen endet (ASCII-Steuerzeichen dezimal 13 und 7). Diese Steuerzeichen sollte man abtrennen, damit der Text in sauberer Form vorliegt.

· Bevor auf eine Tabelle zugegriffen wird, sollte immer überprüft werden, ob die entsprechende Tabelle im Dokument wirklich vorhanden ist, sonst bricht der Code mit einer Fehlermeldung ab. Dasselbe gilt für Zeilen und Spalten oder Zellen.

Feststellen, ob eine entsprechende Tabelle im Dokument vorhanden ist

· Zurück zum Anfang
Ermitteln, ob das Dokument Tabellen enthält

Im Folgenden wird ermittelt, ob sich im Hauptteil des Dokuments eine Tabelle befindet. Wenn nicht wird die Verarbeitung abgebrochen.



Sub PruefenObTabelleImHauptteilVorhanden()

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Der Hauptteil des Dokuments enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  ActiveDocument.Tables(1).Select

End Sub
(Überprüft, ob der Hauptteil des Dokuments eine Tabelle enthält.)



Im Folgenden wird ermittelt, ob die primäre Kopfzeile des ersten Abschnittes im Dokument, eine Tabelle enthält:



Sub PruefenObTabelleInDerPrimaerenKopfzeileVorhanden()

  With ActiveDocument.Sections(1).Headers(wdHeaderFooterPrimary).Range

    If .Tables.Count = 0 Then
      MsgBox "Die primäre Kopfzeile des 1. Abschnittes enthält keine Tabelle."

      Exit Sub
    End If
    .Tables(1).Select

  End With
End Sub
(Überprüft, ob die primäre Kopfzeile des ersten Abschnittes im Dokument eine Tabelle enthält.)



Ermitteln, ob sich die Einfügemarke in einer Tabelle befindet

Im Folgenden wird ermittelt, ob sich die Einfügemarke in einer Tabelle befindet. Wenn nicht wird die Verarbeitung abgebrochen.



Sub PruefenObSichDieEinfuegemarkeInEinerTabelleBefindet()

  If Selection.Information(wdWithInTable) = False Then
    MsgBox "Die Einfügemarke befindet sich nicht in einer Tabelle.", vbInformation

    Exit Sub
  End If
  Selection.Tables(1).Select

End Sub
(Überprüft, ob die Einfügemarke in einer Tabelle steht.)



Text aus einer Tabellenzelle auslesen

· Zurück zum Anfang
Wie schon Eingangs erwähnt, können Sie eine Zelle direkt über das Tabellenobjekt oder aber über die Zeile bzw. die Spalten, welche diese Zelle beherbergt, auslesen.

Beachten Sie, dass der Text in der Zelle immer mit zwei Steuerzeichen endet, welche abgetrennt werden sollten.

Im Folgenden wird der Text der Zelle, welche sich in der 3. Zeile und in der 2. Spalte befindet (man könnte diese Zelle gedanklich auch mit B3 adressieren) ausgelesen. Sie finden nun alle drei Methoden, um diese Problemstellung anzugehen.



Sub TabellenzelleAuslesen1()

  strText = ActiveDocument.Tables(1).Cell(3, 2).Range.Text

  strText = Left(strText, Len(strText) - 2)

  MsgBox strText

End Sub
(Die Zelle mit der Adresse «B3» wird ausgelesen. Dabei wird die Zelle direkt als Element der Tabelle angesprochen.)





Sub TabellenzelleAuslesen2()

  strText = ActiveDocument.Tables(1).Rows(3).Cells(2).Range.Text

  strText = Left(strText, Len(strText) - 2)

  MsgBox strText

End Sub
(Die Zelle mit der Adresse «B3» wird ausgelesen. Dabei wird diese Zelle als Element der dritten Tabellenzeile angesprochen.)





Sub TabellenzelleAuslesen3()

  strText = ActiveDocument.Tables(1).Columns(2).Cells(3).Range.Text

  strText = Left(strText, Len(strText) - 2)

  MsgBox strText

End Sub
(Die Zelle mit der Adresse «B3» wird ausgelesen. Dabei wird die Zelle als Element der zweiten Tabellenspalte angesprochen.)



Sollte die Tabelle numerische Werte (als Zahlen) enthalten, was bei Tabellen in der Praxis oft vorkommt, dann können die Steuerzeichen auch unter Einsatz der VB(A) Val-Funktion eliminiert werden. Dabei trifft man dann gleich zwei Fliegen auf einen Schlag. Einerseits wird man die störenden Steuerzeichen los und anderseits erhält man eine Variable des Typus Double zurück mit welcher direkt Berechnungen ausgeführt werden können.



Sub TabellenzelleAuslesen4()

  Ausgaben = Val(ActiveDocument.Tables(1).Columns(2).Cells(3).Range.Text)

  MsgBox Ausgaben

End Sub
(Enthält die Tabellenzelle einen numerischen Wert, dann kann auch die «VAL-Funktion» dafür herhalten, die störenden Sonderzeichen, welche einen Tabellenzelle abschliessen, zu eliminieren.)



Anmerkung

· Sollte die Tabellenzelle überhaupt keine Zahl enthalten oder leer sein, dann gibt die Val-Funktion den Wert 0 zurück.

Ganz allgemein: um sicher zu gehen, dass der Code nicht mit einer Fehlermeldung abbricht, müssten Sie eigentlich noch prüfen, ob die Tabelle und die zu adressierende Zelle wirklich vorhanden sind. Im Folgenden sehen Sie den Code des ersten Beispiels in diesem Kapitel nochmals, diesmal aber mit der erwähnten Fehlerbehandlung.



Sub TabellenzelleAuslesenMitFehlerbehandlung()

  Dim oTable As Table

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Das Dokument enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables(1)

  If oTable.Rows.Count < 3 Or oTable.Columns.Count < 2 Then
    MsgBox "Die angeforderte Zelle existiert nicht.", vbInformation

    Exit Sub
  End If
  strText = ActiveDocument.Tables(1).Cell(3, 2).Range.Text

  strText = Left(strText, Len(strText) - 2)

  MsgBox strText

End Sub
(Zuerst wird sichergestellt, dass die angeforderte Tabelle und Zelle tatsächlich existiert. Anschliessend wird der Text der Zelle «B3» ausgelesen.)



Word-Tabelle in einem Datenfeld (Array) abbilden

· Zurück zum Anfang
Eine Word-Tabelle lässt sich sehr einfach in einem Datenfeld (Array) abbilden. Am besten eignet sich dazu ist ein zweidimensionaler Array. Dabei stellt die erste Dimension die einzelnen Tabellenzeilen und die zweite Dimension die einzelnen Tabellenspalten dar. Selbstverständlich ist das auch umgekehrt möglich.

Im Folgenden finden Sie zuerst eine allgemeingültige Routine um eine Tabelle in einen eindimensionalen Array zu laden. Dabei wird mit der Zelle oben links in der Tabelle begonnen. Anschliessend folgen die nächsten Zellen in Leserichtung, also von links nach rechts und dann von oben nach unten. Das Beispiel spricht die erste Tabelle im Hauptteil des Dokuments an.

Eindimensionales Datenfeld (Array)



Sub WordTabelleNachEindimensionalenArray1()

  Dim x() As String
  Dim oTable As Table, Zelle As Cell

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Das Dokument enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables(1)

  ReDim x(1 To oTable.Range.Cells.Count)

  i = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
  For Each Zelle In oTable.Range.Cells

    i = i + 1

    strText = Zelle.Range.Text

    x(i) = Left(strText, Len(strText) - 2)

  Next Zelle

End Sub
(Eine Word-Tabelle wird in einen eindimensionalen Array geladen. Die Leserichtung in diesem Beispiel lautet: A1, B1, C1, A2, B2, C2, A3, B3, C3, etc.)



Im Folgenden Beispiel wird nochmals ein eindimensionaler Array der ersten Tabelle im Hauptteil des Dokuments gebildet. Dieses Mal wird die Reihenfolge der Zellen aber anders eingelesen, nämlich von oben nach unten und dann von rechts nach links, also nach dem Muster A1, A2, A3, B1, B2, B3, C1, C2, C3, etc. Das Beispiel spricht die erste Tabelle im Hauptteil des Dokuments an.



Sub WordTabelleNachEindimensionalenArray2()

  Dim x() As String
  Dim oTable As Table, Spalte As Column, Zelle As Cell

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Das Dokument enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables(1)

  ReDim x(1 To oTable.Range.Cells.Count)

  i = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
  For Each Spalte In oTable.Columns

    For Each Zelle In Spalte.Cells

      i = i + 1

      strText = Zelle.Range.Text

      x(i) = Left(strText, Len(strText) - 2)

    Next Zelle

  Next Spalte

End Sub
(Eine Word-Tabelle wird nach einem eindimensionalen Array ausgelesen. In diesem Beispiel werden die Zellen in folgender Reihenfolge ausgelesen: A1, A2, A3, B1, B2, B3, C1, C2, C3, etc.)



Zweidimensionales Datenfeld (Array)

Geeigneter als ein eindimensionales Datenfeld um eine Tabelle auszulesen, ist ein zweidimensionaler Array. Im folgenden Beispiel wird eine allgemeingültige Routine aufgezeigt, welche diese Problemstellung löst.

Im ersten Beispiel stellt die erste Dimension die Zeilen und die zweite Dimension die Spalten dar. Im zweiten Beispiel wird die Tabelle dann transponiert, d.h. die Spalten stellen die erste Dimension und die Zeilen die zweite Dimension dar. Die Beispiele sprechen die erste Tabelle im Hauptteil des Dokuments an.



Sub WordTabelleNachZweidimensionalenArray1()

  Dim x() As String
  Dim oTable As Table, Zeile As Row, Zelle As Cell

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Das Dokument enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables(1)

  ReDim x(1 To oTable.Rows.Count, 1 To oTable.Columns.Count)

  i = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
  For Each Zeile In oTable.Rows

    i = i + 1

    j = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
    For Each Zelle In Zeile.Cells

      j = j + 1

      strText = Zelle.Range.Text

      x(i, j) = Left(strText, Len(strText) - 2)

    Next Zelle

  Next Zeile

End Sub
(Eine Word-Tabelle wird in nach einem zweidimensionalen Array ausgelesen. In diesem Beispiel stellen die Zeilen die erste Dimension und die Spalten die zweite Dimension dar.)



Nun wird die Tabelle transponiert, so dass die Spalten die erste Dimension und die Zeilen die zweite Dimension darstellen.



Sub WordTabelleNachZweidimensionalenArray2()

  Dim x() As String
  Dim oTable As Table, Spalte As Column, Zelle As Cell

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Das Dokument enthält keine Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables(1)

  ReDim x(1 To oTable.Columns.Count, 1 To oTable.Rows.Count)

  i = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
  For Each Spalte In oTable.Columns

    i = i + 1

    j = 0 'Oder auch -1, wenn der Array bei "0" beginnen soll
    For Each Zelle In Spalte.Cells

      j = j + 1

      strText = Zelle.Range.Text

      x(i, j) = Left(strText, Len(strText) - 2)

    Next Zelle

  Next Spalte

End Sub
(Eine Word-Tabelle wird in nach einem zweidimensionalen Array ausgelesen. In diesem Beispiel stellen die Spalten die erste Dimension und die Zeilen die zweite Dimension dar.)



Feststellen in welcher Zelle sich die Einfügemarke befindet

· Zurück zum Anfang
Im Folgenden finden Sie die Mimik um festzustellen, in welcher Zelle sich die Einfügemarke befindet. Ist ein Zellbereich markiert, dann wird die letzte Zelle in der Markierung zurückgegeben.



Sub FeststellenInWelcherZelleSichDieEinfügemarkeBefindet()

  If Selection.Information(wdWithInTable) = False Then
    MsgBox "Die Einfügemarke befindet sich nicht in einer Tabelle.", vbInformation

    Exit Sub
  End If
  x = Selection.Information(wdEndOfRangeRowNumber)

  y = Selection.Information(wdEndOfRangeColumnNumber)

  MsgBox "Die Einfügemarke befindet sich:" & vbCrLf & _

     "in der Zeile: " & x & vbCrLf & "und in der Spalte: " & y

End Sub
(Feststellen in welcher Zelle sich die Einfügemarke befindet.)



Adresse einer Zelle oder eines Zellbereiches ausweisen

· Zurück zum Anfang
Es geht auch etwas schöner, wie das folgende Beispiel beweist. Die folgende Routine benutzt die Notation, welche wir aus Excel kennen, um die "aktive" Zelle, bzw. den markierten Zellbereich zurückzugeben. Die Routine erweist sich im Umgang mit Word-Tabellen immer wieder als ausserordentlich praktisch. Die Zellbezeichnung bzw. der Zellbereich wird nach der Ausführung des Makros in der Statuszeile angezeigt.

· So bringen Sie das Makro ein und verknüpfen es mit einem neuen Schaltknopf auf der Symbolleiste Ihrer Wahl


Private Const Titel = "Tabellen Position"

Sub TabellenPositionAnzeigen()

   If Selection.Information(wdWithInTable) = False Then
      Markierung = 0

      If Selection.Type = 1 Then
        Meldung = "Der Zeiger befindet sich nicht in einer Tabelle!"

      Else
        Meldung = "Die Markierung stellt keinen gültigen Tabellenbereich dar!"

      End If
   Else
      If Selection.Type = 1 Then
         Position = TabellenPosition

         Markierung = 1

         tmp = Position

         Meldung = "Die aktive Zelle heisst: " & tmp & "."

      Else
         Bereich = TabellenBereich

         Markierung = 2

         tmp = Bereich

         Meldung = "Der markierte Tabellenbereich heisst: " & tmp & "."

      End If
   End If
   'MsgBox Meldung, vbInformation, Titel
   StatusBar = Meldung

End Sub
Private Function TabellenPosition() As String
  TabellenPosition = _

     AlphaNum(Selection.Information(wdStartOfRangeColumnNumber)) & _

        Selection.Information(wdStartOfRangeRowNumber)

End Function
Private Function TabellenBereich() As String
   TabellenBereich = _

      AlphaNum(Selection.Information(wdStartOfRangeColumnNumber)) & _

         Selection.Information(wdStartOfRangeRowNumber) & " - " & _

            AlphaNum(Selection.Information(wdEndOfRangeColumnNumber)) & _

               Selection.Information(wdEndOfRangeRowNumber)

End Function
Private Function AlphaNum(Num As Long) As String
  Z = 0

  While Num > 26

    Z = Z + 1

    Num = Num - 26

  Wend
  AlphaNum = Chr(Num + 64)

  If Z > 0 Then AlphaNum = Chr(Z + 64) & AlphaNum

End Function
(Dieser Code teilt dem Benutzer mit, in welcher Zelle sich die Einfügemarke befindet. Sind mehrere Zellen markiert, wird die Start- und Endzelle ausgewiesen.)



Eine bestimmte Tabelle über eine Textmarke auffinden

· Zurück zum Anfang
Das Auffinden einer Tabelle über deren Index erweist sich nicht immer als praxisgerecht, da sich dieser Index ja sehr schnell ändern kann, wenn das Dokument mit zusätzlichen Tabellen bereichert wird bzw. Tabellen gelöscht werden.

Wie schon Eingangs erwähnt, besitzt eine Tabelle keinen (eindeutigen) Namen. Es ist auch nicht möglich einen solchen zu vergeben, auch nicht unter Einsatz von VBA.

Der Ausweg aus diesem Dilemma ist eine Textmarke, welche Sie an beliebiger Stelle in der Tabelle vergeben können. Es ist dann sehr einfach mit VBA auf genau diese Tabelle zuzugreifen.

Angenommen es gibt eine Tabelle, welche eine Textmarke mit dem eindeutigen Namen Jahresabrechnung aufweist, dann können Sie diese Tabelle wie folgt ansprechen:



Sub TabelleUeberTextmarkeAnsprechen()

  Dim oTable As Table

  If ActiveDocument.Bookmarks.Exists("Jahresabrechnung") Then
    Set oTable = _

       ActiveDocument.Bookmarks("Jahresabrechnung").Range.Tables(1)

    oTable.Select

  End If
End Sub
(Eine Tabelle wird über eine Textmarke, welche sich in der entsprechenden Tabelle befindet angesprochen.)



Auf eine Tabelle in der Kopf- oder Fusszeile bzw. auf eine Tabelle in einem Textfeld zugreifen

· Zurück zum Anfang
Bei den bisher aufgeführten Beispielen wurde immer davon ausgegangen, dass sich die Tabelle, welche angesprochen werden soll, im Hauptteil des Dokuments bzw. bei der Einfügemarke befindet. Sie können aber auch auf eine Tabelle ausserhalb des Hauptteils zugreifen und zwar ohne die Ansicht zu wechseln und auch ohne auf das Selection-Objekt angewiesen zu sein.

Tabelle in der Kopf- oder Fusszeile ansprechen

Im folgenden Beispiel wird die erste Zelle der ersten Tabelle, welche sich in der primären Kopfzeile des ersten Abschnittes des Dokuments befindet, ausgelesen:



Sub ErsteTabelleInDerPrimarenKopfzeileImErstenAbschnitDesDokumentsAuslesen()

  Dim oStory As Range, oTable As Table

  Set oStory = ActiveDocument.Sections(1).Headers(wdHeaderFooterPrimary).Range

  Set oTable = oStory.Tables(1)

  strText = oTable.Cell(1, 1).Range.Text

  MsgBox Left(strText, Len(strText) - 2)

End Sub

(Zuerst wird der erste Abschnitt im Dokument und dort die primäre Kopfzeile angepeilt. Dann wird die erste Tabelle in dieser Kopfzeile angesprochen. Schliesslich wird der Inhalt der ersten Zelle in dieser Tabelle ausgelesen.)



Tabelle in einem Textfeld ansprechen

Etwas schlechter sieht es aus, wenn Sie auf eine Tabelle in einem Textfeld zugreifen müssen. Wenn Sie dieses Textfeld ebenfalls mittels VBA-Code erstellt haben, ist die Not allerdings klein, denn Sie können der Shape, welche einem Textfeld obliegt, einen eindeutigen und einprägsamen Namen vergeben. Schwieriger wird es, wenn MS Word den Namen von sich aus gewählt hat. Bedenken Sie in diesem Zusammenhang auch, dass nicht jede Shape zwingend ein Textfeld sein muss. Textfelder stellen hier nur eine Untermenge dar.

Im folgenden Beispiel wird auf die erste Tabelle, welche sich im ersten Textfeld im Hauptteil des Dokuments befindet, zugegriffen:



Sub TabelleInTextfeldAnsprechen()

  Dim oStory As Range

  Set oStory = ActiveDocument.Shapes(1).TextFrame.TextRange

  Set oTable = oStory.Tables(1)

  strText = oTable.Cell(1, 1).Range.Text

  MsgBox Left(strText, Len(strText) - 2)

End Sub
(Eine Tabelle in einem Textfeld wird angesprochen.)



Anmerkungen

· Mehr über den Umgang mit Dokumentteilen, inklusive einer Tabelle aller elf Stories, welche MS Word definiert, finden Sie im folgenden Skript: Das StoryRange-Objekt.

· Ausführliche Informationen über den VBA-technischen Umgang mit Shapes stehen Ihnen im folgenden Skript zur Verfügung: VBA-technischer Umgang mit "Shapes" und "InlineShapes".

Eine Tabelle mit VBA erstellen

· Zurück zum Anfang
Das Erstellen einer Tabelle mittels VBA ist denkbar einfach, der Vorgang lässt sich in der Regel auch problemlos aufzeichnen. Im Folgenden wird an der Stelle der Einfügemarke eine kleine Tabelle erstellt.



Sub TabelleEinfügen1()

  With Selection

    If .Information(wdWithInTable) = True Then
      .MoveDown Count:=.Tables(1).Rows.Count - _

         .Information(wdEndOfRangeRowNumber) + 2

    End If
    ActiveDocument.Tables.Add Range:=.Range, NumRows:=2, NumColumns:=2

  End With
End Sub
(Eine Tabelle wird in das Word-Dokument eingebracht. Sollte sich an der Stelle, wo die Einfügemarke ist, bereits eine Tabelle befinden, wird ausgerechnet, um wie viele Zeilen der Zeiger nach unten hüpfen muss.)



Hier nochmals dieselbe Aufgabenstellung, etwas eleganter gelöst.



Sub TabelleEinfügen2()

  Dim oRange As Range

  Set oRange = Selection.Range

  If Selection.Information(wdWithInTable) = True Then
    oRange.SetRange Start:=Selection.Tables(1).Range.End + 1, _

                    End:=Selection.Tables(1).Range.End + 1

  End If
  ActiveDocument.Tables.Add Range:=oRange, NumRows:=2, NumColumns:=2

End Sub
(Verfolgt das selbe Ziel, wie das obige Beispiel. Diesmal wird aber mit einem Range-Objekt gearbeitet. Diese Lösung ist der obigen vorzuziehen.)



· Beachten Sie, dass zwischen zwei Tabellen im Minimum eine Absatzmarke stehen muss. Folgen sich nämlich zwei Tabellen nahtlos, dann verschmelzen diese zu einer grossen Tabelle.

Im Folgenden wird eine kleine Tabelle erstellt. Anschliessend werden die Zellen der Tabellen mit Text gefüllt.



Sub TabelleErstellen()

  Dim oTable As Table, Zelle As Cell

  If Selection.Information(wdWithInTable) = True Then
    MsgBox "Die Einfügemarke befindet bereits in einer Tabelle.", vbInformation

    Exit Sub
  End If
  Set oTable = ActiveDocument.Tables.Add(NumRows:=2, NumColumns:=2, _

     Range:=Selection.Range)

  i = 0

  For Each Zelle In oTable.Range.Cells

    i = i + 1

    Zelle.Range.Text = "Das ist die " & i & ". Zelle"

  Next
End Sub
(Eine Tabelle wird mit VBA erstellt. Die Tabelle wird am Ort der Einfügemarke generiert. Anschliessend werden die Zellen dieser Tabelle mit Daten gefüllt.)



· Bei grösseren und oder komplexeren Tabellen ist es äusserst ineffizient zuerst eine Tabelle zu erstellen und anschliessend diese Tabelle mit Daten zu füllen. Um Dimensionen schneller ist es, zuerst die Daten zusammenzusetzen und dann diese Daten in eine Tabelle umzuwandeln. Die VBA-Methode mit dem Namen ConvertToTable eignet sich dazu hervorragend und ist einfache anzuwenden, wie das folgende Beispiel aufzeigt.

· Für Word2000 gilt, dass sich die Verarbeitungsgeschwindigkeit von Tabellen mit der Installation des Office Service Releases 1 verbessern soll.

Ziel des folgenden Codes ist es, Schriftproben für alle Schriften zu erstellen, welche unter MS Word zur Verfügung stehen. Dafür wird eine zweispaltige Tabelle erzeugt, wobei in der linken Spalte der Schriftname erscheint und in der rechten Spalte ein Beispieltext, welcher mit der Schriftart in der linken Spalte formatiert ist. Die Routine vollzieht folgende Schritte:

1. Es wird ein neues Dokument erstellt, welches die Tabelle aufnehmen soll.

2. Die Schriften werden zusammengesucht. Abwechslungsweise wird der Schriftname und dann der Beispieltext in einer Zeichenfolge verkettet. Als Trennzeichen für die Spalten dient ein Tabulator. Als Steuerzeichen für eine neue Zeile dient ein Zeilenumbruch.

3. Nun wird noch die Kopfzeile der Tabelle vorne an die Zeichenfolge angehängt.

4. Die so konstruierte Zeichenfolge wird in das Dokument geschrieben.

5. Der entstandene Text wird mittels ConvertToTable-Methode in eine Tabelle umgewandelt.

6. Die rechte Spalte, welche den Beispieltext enthält, wird mit der Schriftart formatiert, welche in der linken Spalte aufgeführt ist.

Kopieren Sie den folgenden Code in ein neues Modul und testen Sie das Makro.

· So bringen Sie das Makro ein und verknüpfen es mit einem neuen Schaltknopf auf der Symbolleiste Ihrer Wahl


Private Const Probetext = _

   "Franz jagt im komplett verwahrlosten Taxi quer durch Bayern. 0123456789"

Sub SchriftProbeErstellen()

  For Each Schrift In Application.FontNames

    strText = strText & Schrift & vbTab & Probetext & vbCrLf

  Next
  Dim nDoc As Document, Zeile As Row

  Set nDoc = Documents.Add

  Application.ScreenUpdating = False
  nDoc.Content = "Schriftart" & vbTab & Probetext & vbCrLf & strText

  nDoc.Content.ConvertToTable Separator:=vbTab, _

     Format:=wdTableFormatProfessional, AutoFit:=True
  For Each Zeile In nDoc.Tables(1).Rows

    If Zeile.Index > 1 Then
      Zeile.Range.Cells(2).Range.Font.Name = _

         Left(Zeile.Cells(1).Range.Text, _

            Len(Zeile.Cells(1).Range.Text) - 2)

      StatusBar = Zeile.Index - 1 & " Schriften formatiert - Bitte warten..."

    End If
  Next
  StatusBar = Application.FontNames.Count & " Schriften verfügbar."

  Application.ScreenUpdating = True
End Sub
(Erstellt eine Schriftprobe für jede Schriftart, welche unter Word zur Verfügung steht. Da das Arbeiten mit grossen Tabellen sehr langsam sein kann, sollten Sie die Tabelle so spät wie möglich generieren.)



Das letzte Beispiel in diesem Kapitel zeigt eines sehr schön: die Routine läuft sehr schnell, solange nicht mit der Tabelle selbst gearbeitet wird. Auch die ConvertToTable-Methode ist ordentlich schnell. Der Schluss, in welchem eine Word-Tabelle nachformatiert wird, geht dann aber schleppend voran.

Es ist also wichtig, möglichst lange ohne Tabelle zu arbeiten. Im Beispiel wird eine String-Variable verwendet, um das Resultat vorzubereiten. Diese Zeichenfolge, welche die gesamte Substanz für die Tabelle enthält, wird dann in das Dokument geschrieben. Erst ganz um Schluss wird dieser Text in eine Tabelle umgewandelt.

Um im Text die zukünftigen Spalten anzudeuten, können Sie das Tab-Steuerzeichen verwenden. Wenn Sie annehmen müssen, dass der Inhalt der Zellen selbst solche Tab-Steuerzeichen enthalten könnte, dann können Sie problemlos auf ein anderes nicht druckbares Zeichen ausweichen. Nicht jedes Zeichen eignet sich dafür. Verwenden Sie z.B. das ASCII-Zeichen dezimal 1 wie im folgenden Beispiel und reflektieren Sie diese Entscheidung in der ConvertToTable-Methode.

Im folgenden Beispiel werden die Dateien eines Verzeichnisses auf Ihrem PC aufgelistet. Die Namen dieser Dateien werden mit dateispezifischen Informationen ergänzend. Es wird dann eine Tabelle erstellt, welche Dateinamen und diese zusätzlichen Informationen sortiert nach Dateigrösse, enthält. Der programmtechnische Ablauf ist wie folgt:

1. Die Dateien eines vorgegebenen Verzeichnisses werden ausgelesen und mit dateispezifischen Informationen ergänzt.

2. In einer String-Variablen wird immer genau eine Zeile zusammengesetzt. Als Trennzeichen (Delimiter) für die Spalten dient das ASCII-Steuerzeichen dezimal 1. Als Zeilenabschluss dient eine Absatzmarke (ASCII-Steuerzeichen dezimal 13). Diese String-Variable wird dann in das Dokument geschrieben.

3. Wenn alle Dateien des Verzeichnisses in genannter Weise abgearbeitet sind, wird den Daten noch die Kopfinformation beigefügt. Nun wird eine Tabelle erstellt. Diese Tabelle wird dann mit der Sort-Methode sortiert.

4. Einzelne Spalten der Tabelle werden nun noch nachformatiert, damit die Tabelle an Lesbarkeit gewinnt.

Beachten Sie, dass die Mimik um die Dateien zusammen zu suchen, nicht rekursiv ist. Es werden also ausschliesslich Dateien gefunden, welche sich direkt im vorgegebenen Verzeichnis befinden. Unterverzeichnisse dagegen, werden nicht durchsucht.

Wenn Sie ein Verzeichnis vorgeben, welche viele Dateien enthält, dann kann sich das Ergebnis über viele Seiten erstrecken. Spätestens dann werden Sie einmal mehr erkennen können, dass das Zusammenstellen der Daten und die Generierung der Tabelle schnell abläuft. Die Nachformatierung aber, die mit dem Tabellenobjekt gemacht werden muss, ist ein schleppender Vorgang.

Die Formatierung der Spalte Dateigrösse muss nachträglich erfolgen, damit die Table-Sort-Methode zu einem korrekten Ergebnis führt. Durch diesen zusätzlichen Vorgang verlängert sich die Laufzeit der Routine um ein zigfaches.

Kopieren Sie den folgenden Code in Ihrer VBA-Umgebung in ein neues Modul. Passen Sie im Kopf des Codes die Konstante mit dem Namen Verzeichnis Ihren Bedürfnissen an. Wenn Sie den Code ablaufen lassen, beobachten Sie die Statuszeile Ihres Word-Fensters. Sie sollten an Hand der Statusinformation feststellen können, welche Etappen beim Aufbau der Tabelle zeitintensiv sind.



Private Const Verzeichnis = "C:\WINNT"

Sub DateienAuflisten()

  Dim nDoc As Document

  If Dir(Verzeichnis & "\") = "" Then
    MsgBox "Das Verzeichnis " & UCase(Verzeichnis) & " existiert nicht."

    Exit Sub
  End If
  Set nDoc = Documents.Add

  Application.ScreenUpdating = False
  'Dateien auslesen
  x = Dir(Verzeichnis & "\", vbArchive + vbHidden + vbReadOnly)

  While Not x = ""

    nDoc.Content.InsertAfter x & Chr(1) & GroesseAuslesen(x) & Chr(1) & _

       DatumAuslesen(x) & Chr(1) & AttributeAuslesen(x) & vbCrLf

    x = Dir

  Wend
  'Schriftgrad und Abstände vorgeben
  nDoc.Content.Font.Size = 9

  nDoc.Content.ParagraphFormat.SpaceBefore = 2

  nDoc.Content.ParagraphFormat.SpaceAfter = 2

  'Kopfinformation der Tabelle im Dokument einfügen
  nDoc.Content.InsertBefore "Dateiname" & Chr(1) & "Grösse" & Chr(1) & _

     "Datum" & Chr(1) & "Schreibgeschützt" & Chr(1) & "Versteckt" & vbCrLf

  'Text in Tabelle konvertieren
  StatusBar = "Die Tabelle wird jetzt erstellt."

  Application.ScreenRefresh

  Dim oTable As Table, Zelle As Cell

  Set oTable = nDoc.Content.ConvertToTable(Separator:=Chr(1), AutoFit:=True)

  'Zeilen nach Dateigrösse sortieren
  oTable.Sort ExcludeHeader:=True, FieldNumber:=2, _

     SortFieldType:=wdSortFieldNumeric, SortOrder:=wdSortOrderDescending

  'Feld Dateigrösse formatieren
  StatusBar = _

     "Die Tabelle wird nun nachformatiert. Das nimmt einige Zeit in Anspruch."

  Application.ScreenRefresh

  Assistant.Animation = msoAnimationWorkingAtSomething

  System.Cursor = wdCursorWait

  For Each Zelle In oTable.Columns(2).Cells

    If Zelle.RowIndex > 1 Then
      tStr = Left(Zelle.Range.Text, Len(Zelle.Range.Text) - 2)

      tStr = Format(tStr, "###'###'##0")

      ofs = InStr(tStr, "'")

      While ofs = 1 'Strip leading quotes
        tStr = Right(tStr, Len(tStr) - 1)

        ofs = InStr(tStr, "'")

      Wend
      Zelle.Range.Text = tStr

      Zelle.Range.Font.Name = "Courier New"

      Zelle.Range.ParagraphFormat.Alignment = wdAlignParagraphRight

    End If
  Next
  'Textausrichtung in Spalten 4 + 5 ändern
  nDoc.Range(Start:=oTable.Cell(2, 4).Range.Start, _

    End:=oTable.Cell(oTable.Rows.Count, 5).Range.End).Select

  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter

  'Kopf der Tabelle auf jeder Seite wiederholen
  oTable.Rows(1).HeadingFormat = True
  'Zum Anfang im Dokument springen
  nDoc.Range(0, 0).Select

  StatusBar = "Die Liste ist nun erstellt."

  System.Cursor = wdCursorNormal

  Assistant.Animation = msoAnimationCharacterSuccessMajor

  Application.ScreenUpdating = True
End Sub
Private Function GroesseAuslesen(ByVal Datei As String) As Long
  GroesseAuslesen = FileLen(Verzeichnis & "\" & Datei)

End Function
Private Function DatumAuslesen(ByVal Datei As String) As Date
  DatumAuslesen = FileDateTime(Verzeichnis & "\" & Datei)

End Function
Private Function AttributeAuslesen(ByVal Datei As String) As String
  Attr = GetAttr(Verzeichnis & "\" & Datei)

  If Attr >= 32 Then Attr = Attr - 32 'Archive
  Select Case Attr

    Case 0: t = "-" & Chr(1) & "-"

    Case 1: t = "X" & Chr(1) & "-"

    Case 2: t = "-" & Chr(1) & "X"

    Case 3: t = "X" & Chr(1) & "X"

  End Select
  AttributeAuslesen = t

End Function
(Eine Word-Tabelle mit umfangreichen Daten wird gänzlich mittels VBA erstellt. Im Gegensatz zum vorhergehenden Beispiel, erfolgt der Aufbau hier Zeile um Zeile, welche fortwährend in das Dokument geschrieben werden. Schliesslich werden diese Zeilen in eine Tabelle umgewandelt. Die Tabelle wird anschliessend noch nachformatiert.)



Eine Tabelle in der Kopf- oder Fusszeile bzw. in einem Textfeld erstellen

· Zurück zum Anfang
Bei den bisher aufgeführten Beispielen wurde immer davon ausgegangen, dass die Tabelle, in den Hauptteil des Dokuments bzw. bei der Stelle, an der sich die Einfügemarke befindet, eingefügt werden soll. Sie können aber auch eine Tabelle ausserhalb des Hauptteils erstellen und zwar ohne die Ansicht zu wechseln und auch ohne auf das Selection-Objekt angewiesen zu sein.

Tabelle in der Kopfzeile erstellen

Im folgenden Beispiel wird eine Tabelle in die primäre Kopfzeile des ersten Abschnittes im Dokument eingefügt:



Sub TabelleInDiePrimaereKopfZeileDesErstenAbschnittesEinfügen()

  Dim oStory As HeaderFooter, oTable As Table

  Set oStory = ActiveDocument.Sections(1).Headers(wdHeaderFooterPrimary)

  Set oTable = oStory.Range.Tables.Add(Range:=oStory.Range, _

     Numrows:=2, NumColumns:=2)

  i = 0

  For Each oCell In oTable.Range.Cells

    i = i + 1

    oCell.Range.Text = CStr(i) & ". Zelle"

  Next
End Sub
(Eine Tabelle wird in die primäre Kopfzeile des ersten Abschnitts im Dokument eingebracht.)



Anmerkung

· Mehr über den Umgang mit Dokumentteilen, inklusive einer Tabelle aller elf Stories, welche MS Word definiert, finden Sie im folgenden Skript: Das StoryRange-Objekt.

Tabelle in einem Textfeld erstellen

Im Folgenden wird eine Tabelle in ein Textfeld, welches mit demselben Code erstellt wird, eingebracht. Ein Textfeld stellt in MS Office letztlich eine Shape dar. Diese Shape wird im Bezug auf den Seitenrand positioniert. Das Resultat ist eine kleine Tabelle am unteren, rechten Blattrand.



Sub TextfeldErstellenUndTabelleInDiesesTextfeldEinbringen()

  Dim oShape As Shape, oTable As Table, oCell As Cell

  Breite = 5
  Hoehe = 2
  AbstandRechts = 2
  AbstandUnten = 2.5
  With ActiveDocument.Sections(1).PageSetup

    x = .PageWidth - CentimetersToPoints(Breite + AbstandRechts)

    y = .PageHeight - CentimetersToPoints(Hoehe + AbstandUnten)

  End With
  Set oShape = _

     ActiveDocument.Shapes.AddTextbox(Orientation:=msoTextOrientationHorizontal, _

                                      Left:=x, _

                                      Top:=y, _

                                      Width:=CentimetersToPoints(Breite), _

                                      Height:=CentimetersToPoints(Hoehe))

  'oShape.Line.Visible = False
  Set oTable = ActiveDocument.Tables.Add(Range:=oShape.TextFrame.TextRange, _

     NumRows:=3, NumColumns:=2)

  i = 0

  For Each oCell In oTable.Range.Cells

    i = i + 1

    oCell.Range.Text = CStr(i) & ". Zelle"

  Next
End Sub
(Zuerst wird ein Textfeld erstellt. Anschliessend wird in dieses Textfeld eine Tabelle eingebracht. Schliesslich werden die Zellen dieser Tabelle mit Text gefüllt.)



Etwas schlechter sieht es aus, wenn Sie eine Tabelle in einem bereits bestehenden Textfeld erstellen müssen. Aus den wahrscheinlich etwa vier Möglichkeiten, welche Ihnen in diesem Fall offen stehen, habe ich die beiden wirklich praxistauglichen herausgegriffen.

Zuerst muss das bereits existierende Textfeld, welches von MS Word ziemlich zufällig benannt wurde einen aussagekräftigen und eindeutigen Namen erhalten.

Textfeld nachträglich benennen

MS Word vergibt die Namen von Shapes intern zwar eindeutig, letztlich aber ziemlich zufällig. Um einer Shape nachträglich einen aussagekräftigen Namen zu verpassen, können Sie das Textfeld markieren und folgenden Code genau einmal ablaufen lassen:



Sub TextfeldBenennen()

  Dim oShape As Shape

  If Selection.ShapeRange.Count = 0 Then
    MsgBox "Es ist keine Textfeld markiert.", vbInformation

    Exit Sub
  End If
  strName = Selection.ShapeRange(1).Name

  Antwort = InputBox("Geben Sie einen Namen, für das Textfeld ein.", , strName)

  If Antwort = "" Then Exit Sub
  Selection.ShapeRange.Name = Antwort

End Sub
(Jedes Textfeld stellt grundsätzlich eine «Shape» dar. Mit dem folgenden Code können Sie die «Shape», welche dem markierten Textfeld obliegt, benennen:



Anmerkung

· Ausführliche Informationen über den VBA-technischen Umgang mit Shapes stehen Ihnen im folgenden Skript zur Verfügung: VBA-technischer Umgang mit "Shapes" und "InlineShapes".

Nun können Sie dieses Textfeld wie folgt ansprechen und darin eine Tabelle erstellen:



Sub TextfeldAufGrundSeinesNamensAuffindenUndDarinEineTabelleErstellen()

  TFName = "Schlussbilanz"

  Dim oShape As Shape, oTable As Table

  For Each oShape In ActiveDocument.Shapes

    If oShape.Name = TFName Then
      Flag = True
      Set oTable = ActiveDocument.Tables.Add(Range:=oShape.TextFrame.TextRange, _

         Numrows:=2, NumColumns:=2)

      i = 0

      For Each oCell In oTable.Range.Cells

        i = i + 1

        oCell.Range.Text = CStr(i) & ". Zelle"

      Next
    End If
  Next
  If Flag = False Then
    Meldung = "Es existiert keine Textfeld mit dem Namen " & TFName

    Meldung = Meldung & " im Haupteil des Dokuments."

    MsgBox Meldung

    Exit Sub
  End If
End Sub
(Erstellt eine Tabelle in einem benannten Textfeld.)



Ein eleganter Ansatz ist es, ein Textfeld auf Grund einer eindeutigen Textmarke, welche sich an beliebiger Stelle im Textfeld befinden kann, aufzufinden. Bei dieser Mimik gibt es kaum Abhängigkeiten, welche berücksichtigt werden müssen. Letztlich muss nur die nominierte Textmarke existieren und es ist Wurst:

· ob sich die Textmarke im Hauptteil, in einem Textfeld oder in der Kopf-/Fusszeile befindet.

· ob die Textmarke in einem Textfeld, welches sich in der Kopf-/Fusszeile befindet, steht.

· in welcher Ansicht Sie sich gerade befinden.

Das folgende Beispiel setzt einzig voraus, dass eine Textmarke mit der Bezeichnung Schlussbilanz existiert:



Sub TextfeldAufGrundEinerTextmarkeAuffindenUndDarinEineTabelleErstellen()

  TMName = "Schlussbilanz"

  Dim oTable As Table

  If Not ActiveDocument.Bookmarks.Exists(TMName) Then
    MsgBox "Es existiert keine Textmarke mit dem Namen " & TMName & "."

    Exit Sub
  End If
  Set oTable = _

     ActiveDocument.Tables.Add(Range:=ActiveDocument.Bookmarks(TMName).Range, _

        Numrows:=2, NumColumns:=2)

  i = 0

  For Each oCell In oTable.Range.Cells

    i = i + 1

    oCell.Range.Text = CStr(i) & ". Zelle"

  Next
End Sub
(Erstellt eine Tabelle und zwar am Ort an dem sich die Textmarke mit der Bezeichnung «Schlussbilanz» befindet. Dabei ist es gleichgültig, ob sich diese Textmarke im Hauptteil des Dokuments oder aber in einem Textfeld befindet.)



Anmerkungen

· Der VBA-technischen Umgang mit Textmarken ist Thema des folgenden Skripts VBA-technischer Umgang mit Textmarken.

· Ausführliche Informationen über den VBA-technischen Umgang mit Shapes stehen Ihnen im folgenden Skript zur Verfügung: VBA-technischer Umgang mit "Shapes" und "InlineShapes".

Zusätzliche Zeilen oder Spalten in eine bestehende Tabelle einbringen

· Zurück zum Anfang
Um eine bestehende Tabelle zu vergrössern können Sie wie folgt vorgehen:

Zusätzliche Zeilen einbringen

Das folgende Beispiel ergänzend eine bestehend Tabelle - am Ende - mit einer zusätzlichen Zeile.



Sub TabelleUntenUmEineZeileErweitern()

  Dim oTable As Table, Zeile As Row

  Set oTable = ActiveDocument.Tables(1)

  Set Zeile = oTable.Rows.Add

End Sub
(Eine zusätzliche Zeile wird unten an eine Tabelle angehängt.)



Im folgenden Beispiel wird die Tabelle um eine Zeile erweitert. Diese Zeile wird am Anfang der Tabelle eingefügt.



Sub TabelleObenUmEineZeileErweitern()

  Dim oTable As Table, Zeile As Row

  Set oTable = ActiveDocument.Tables(1)

  Set Zeile = oTable.Rows.Add(BeforeRow:=oTable.Rows(1))

End Sub
(Eine zusätzliche Zeile wird oben an eine Tabelle angehängt.)



Im folgenden Beispiel wird eine Tabelle am Ende mit einer Spalte erweitert. Zusätzlich wird in die erste Zelle dieser neuen Spalte eine Textmarke eingebracht.



Sub TabelleUntenUmEineZeileErweiternUndTextmarkeEinfuegen()

  Dim oTable As Table, Zeile As Row

  Set oTable = ActiveDocument.Tables(1)

  Set Zeile = oTable.Rows.Add

  ActiveDocument.Bookmarks.Add Name:="Kurs" & Format(Zeile.Index, "000"), _

     Range:=ActiveDocument.Range(Zeile.Range.Start, Zeile.Range.Start)

End Sub
(Eine zusätzliche Zeile wird unten an eine Tabelle angehängt. In die erste Zelle dieser neuen Zeile wird dann eine Textmarke eingefügt.) 



Zusätzliche Spalten einbringen

Es folgen nun noch zwei analoge Beispiele diesmal aber für Spalten.

Tabelle rechts um eine Spalte erweitern:



Sub TabelleRechtsUmEineSpalteErweitern()

  Dim oTable As Table, Spalte As Column

  Set oTable = ActiveDocument.Tables(1)

  Set Spalte = oTable.Columns.Add

End Sub
(Eine zusätzliche Spalte wird rechts an eine Tabelle angehängt.)



Tabelle links um eine Spalte erweitern:



Sub TabelleLinksUmEineSpalteErweitern()

  Dim oTable As Table, Spalte As Column

  Set oTable = ActiveDocument.Tables(1)

  Set Spalte = oTable.Columns.Add(BeforeColumn:=oTable.Columns(1))

End Sub
(Eine zusätzliche Spalte wird links an eine Tabelle angehängt.)



Arbeiten mit Tabellenspalten

· Zurück zum Anfang
Im folgenden Beispiel wird in der dritten Spalte der ersten Tabelle im Hauptteil des Dokuments der Ortsbegriff Bâle durch das französischsprachige Äquivalent Bâle ersetzt. Da die Zellen einer Spalte - betrachtet man es in der üblichen Leserichtung - keinen zusammenhängenden Bereich darstellen, behandelt das folgende Beispiel die Spalten als eine Sammlung von Zellen, welche mit einer Schleife bedient werden können. Das Objekt Column gibt es zwar im Zusammenhang mit Word-Tabellen, allerdings verfügt es über keine Range-Eigenschaft. Diese Tatsache erstaunt nicht, denn eine Spalte hat im Sinne des Word-Objekt-Models kein klarer Start- bzw. Endpunkt; sie liegt quer im Felde. Würde man die Spalte in Leserichtung anmalen, müsste der Pinsel immer wieder von neuem angesetzt bzw. abgesetzt werden.

Eine Spalte nach einem bestimmten Text durchsuchen



Sub TabellenspalteDurchsuchen()

  Application.ScreenUpdating = False
  Dim oTable As Table, Zelle As Cell

  Set oTable = ActiveDocument.Tables(1)

  For Each Zelle In oTable.Columns(3).Cells

    With Zelle.Range.Find

      .Text = "Bâle"

      .Replacement.Text = "Bâle"

      .Execute Replace:=wdReplaceAll

    End With
  Next
  Application.ScreenUpdating = True
End Sub
(Diese Routine demonstriert, wie eine Tabellenspalte durchsucht werden kann.)



Spaltenbreite festlegen

Beim Festlegen der Spaltenbreite zeigt die Erfahrung, dass es für ein Gelingen wichtig ist, dass mit der letzten Spalte begonnen wird. Das folgende Beispiel geht von einer vierspaltigen Tabelle aus:



Sub SpaltenBreiteFestlegen()

  Dim oTable As Table, Spalte As Column

  Set oTable = ActiveDocument.Tables(1)

  AnzSpalten = oTable.Columns.Count

  For i = AnzSpalten To 1 Step -1

    Select Case i

      Case 1: oTable.Columns(1).Width = CentimetersToPoints(2)

      Case 2: oTable.Columns(2).Width = CentimetersToPoints(5)

      Case 3: oTable.Columns(3).Width = CentimetersToPoints(2.5)

      Case 4: oTable.Columns(4).Width = CentimetersToPoints(1)

    End Select
  Next i

End Sub
(Für die erste Tabelle im Hauptteil des Dokuments wird die Breite für die vier Tabellenspalten festgelegt.)



Die Range-Eigenschaft einer Tabelle ansprechen

· Zurück zum Anfang
Kapitelüberschrift vor eine Tabelle einfügen

Im folgenden Beispiel soll vor einer bestehenden Tabelle (die erste im Hauptteil des Dokuments) eine Kapitelüberschrift eingefügt werden. Das ist gar nicht so einfach zu lösen, speziell dann nicht, wenn das Dokument unmittelbar mit einer Tabelle beginnt.

Der folgende Code ist allgemeingültig. Er löst diese Aufgabe auch dann, wenn die Tabelle das Erste ist, was im Dokument steht und damit erst Platz am Anfang geschaffen werden muss.



Sub VorDerTabelleEineUeberschriftSchreiben()

  Dim oTable As Table, oRange As Range, sRange As Range

  Set oTable = ActiveDocument.Tables(1)

  Set oRange = oTable.Range

  If oRange.Start = 0 Then
    Set sRange = Selection.Range

    ActiveDocument.Range(0, 0).Select

    Selection.SplitTable

    Selection.Range = sRange

  End If
  oRange.SetRange Start:=oRange.Start - 1, End:=oRange.Start - 1

  oRange.InsertBefore "Periodensystem der chemischen Elemente"

  oRange.Style = wdStyleHeading3

End Sub
(Vor der ersten Tabelle im Dokument, wird eine Kapitelüberschrift gesetzt.)



Beschriftung hinter eine Tabelle einfügen

Im Folgenden wird hinter einer bestehenden Tabelle (die erste im Hauptteil des Dokuments) ein Beschriftung eingefügt. Manuell würden Sie dies über den Word-Dialog Einfügen - Beschriftung… bewerkstelligen:



Sub NachDerTabelleEineBeschriftungSchreiben()

  Dim oTable As Table, oRange As Range

  Set oTable = ActiveDocument.Tables(1)

  Set oRange = oTable.Range

  oRange.SetRange Start:=oRange.End, End:=oRange.End

  oRange.InsertCaption Label:="Tabelle"

End Sub

(Hinter der ersten Tabelle im Hauptteil des Dokuments wird eine Beschriftung eingefügt.)



Höhe einer Tabelle bestimmen

Im folgenden Codebeispiel wird die Höhe (in Millimeter) einer Tabelle (die erste im Hauptteil des Dokuments) bestimmt. Dabei wird vorausgesetzt, dass sich dieselbe Tabelle auf ein und der selben Seite befindet.

Damit der Code keine unsinnigen Resultate liefert, sollte die Tabelle in der Seitenlayout-Ansicht angezeigt werden.



Sub HoeheEinerTabelleBestimmen()

  Dim oRange As Range

  Set oRange = ActiveDocument.Tables(1).Range

  Hoehe = PointsToMillimeters(ActiveDocument.Range(oRange.End, _

     oRange.End).Information(wdVerticalPositionRelativeToPage) - _

          ActiveDocument.Range(oRange.Start, _

             oRange.Start).Information(wdVerticalPositionRelativeToPage))

  strT = "Die Höhe der Tabelle beträgt " & vbCrLf & Format(Hoehe, "###.0") & " mm"

  MsgBox strT

End Sub
(Die Höhe der ersten Tabelle im Hauptteil des Dokuments wird bestimmt.)



Zellbereich formatieren bzw. markieren

· Zurück zum Anfang
Mitunter taucht die Frage auf, wie man einen ganz bestimmten Zellbereich in einer Word-Tabelle markieren kann, um dann diesen Bereicht auf eine bestimmte Art und Weise zu formatieren.

Um dies zu erreichen, ist es wohl das Geschickteste zuerst einen Range zu definieren. Da jeder Zellinhalt selbst als ein Range angesehen werden kann, ist es möglich einen Bereich zu definieren, welcher zusammenhängende Zellen beinhaltet.

Im folgenden Beispiel soll der Text in den Spalten 2+3 rechtsbündig ausgerichtet werden. Die erste Zeile, welche die Kopfinformation der Tabelle darstellt, soll davon aber nicht betroffen sein. Konkret soll also der Zellbereich B2 - Cx markiert werden, wobei x für die letzte Zeile der Tabelle steht.



Sub ZellbereichMarkieren()

  Dim oTable As Table, oRange As Range

  Set oTable = ActiveDocument.Tables(1)

  Set oRange = ActiveDocument.Range(Start:=oTable.Cell(2, 2).Range.Start, _

     End:=oTable.Cell(oTable.Rows.Count, 3).Range.End)

  oRange.Select

  Selection.ParagraphFormat.Alignment = wdAlignParagraphRight

  Selection.Collapse

End Sub
(Ein zusammenhängender Zellbereich in einer Tabelle wird markiert und nachformatiert.)



Wenn Das Arbeiten mit dem Selection-Objekt vermieden werden muss, dann bildet sich folgende Variante an:



Sub ZellbereichMarkierenOhneSelect()

  Dim oTable As Table, oCell As Cell

  Set oTable = ActiveDocument.Tables(1)

  For Each oCell In oTable.Range.Cells

    If oCell.RowIndex > 1 Then
      Select Case oCell.ColumnIndex

        Case 2, 3

          oCell.Range.ParagraphFormat.Alignment = wdAlignParagraphRight

      End Select
    End If
  Next
End Sub
(Löst dieselbe Aufgabe wie das oben abgedruckte Beispiel, allerdings unter Umgehung der «Select-Methode».)



Werte in Zellen formatieren

· Zurück zum Anfang
Im Folgenden finden Sie ein Beispiel, wie sich Zellen in einer Tabelle sequentiell abarbeiten lassen. Sobald die Routine einen numerischen Wert in einer Tabelle entdeckt, wird dieser Wert als Geldbetrag formatiert.



Sub WerteInZellenFormatieren()

  Dim oTable As Table, Zelle As Cell

  Set oTable = ActiveDocument.Tables(1)

  For Each Zelle In oTable.Range.Cells

    Betrag = Left(Zelle.Range.Text, Len(Zelle.Range.Text) - 2)

    If IsNumeric(Betrag) Then
      Betrag = Format(Betrag, "###'###'##0.00")

      ofs = InStr(Betrag, "'")

      While ofs = 1 'Strip leading Quotes
        Betrag = Mid(Betrag, 2)

        ofs = InStr(Betrag, "'")

      Wend
      Zelle.Range.Text = "EUR " & Betrag

      Zelle.Range.ParagraphFormat.Alignment = wdAlignParagraphRight

    End If
  Next
End Sub
(Arbeitet Tabellenzellen sequentiell ab und wandelt - sollte der Inhalt numerisch sein - Werte in Geldbeträge um.)



Letzte Tabellenzeile jeder Seite auffinden

· Zurück zum Anfang
Mitunter muss die letzte Zeile auf jeder Seite einer Tabelle, welche sich über mehrere Seiten erstreckt, ermittelt werden, um z.B. ein Zwischentotal beizufügen. Das folgende Beispiel zeigt die Mimik um dies zu realisieren. Dieser Ansatz ist allerdings nicht sehr schnell.



Sub LetzeZeileAufSeiteFinden()

  savEnv = Options.Pagination

  Options.Pagination = False
  ActiveDocument.Repaginate

  Dim oTable As Table, Zeile As Row

  Set oTable = ActiveDocument.Tables(1)

  For Each Zeile In oTable.Rows

    If Zeile.IsLast Then
      Zeile.Cells(1).Range.Text = "Das ist die allerletzte Zeile."

    Else
      If Not Zeile.Range.Information(wdActiveEndPageNumber) = _

         Zeile.Next.Range.Information(wdActiveEndPageNumber) Then
            Zeile.Next.Cells(1).Range.Text = "Das ist die letzte Zeile."

      End If
    End If
  Next
  Options.Pagination = savEnv

End Sub
(Diese Routine findet jeweils die unterste Zeile auf einer Seite und dies für eine Tabelle, welche sich über mehrere Seiten erstreckt. Das Beispiel geht davon aus, dass es sich um die erste Tabelle im Hauptteil des Dokuments handelt.)



Formel in eine Zelle der letzen Tabellenzeile einfügen

· Zurück zum Anfang
Im Folgenden wird in der letzen Zeile (1. Spalte) die Formel Summe(Über) Word 97 bzw. Sum(Über) ab Word2000 eingefügt. Zur Sicherheit werden auch die Felder in der Tabelle aktualisiert.

Kopieren Sie den folgenden Code im VBA-Editor in ein neues Modul.



Private Const Ausdruck = "=SUM(Über)"

Sub FormelInLetzeTabellenSpalte1()

  Dim oTable As Table, Zelle As Cell

  Set oTable = ActiveDocument.Tables(1)

  oTable.Cell(oTable.Rows.Count, 1).Formula Ausdruck

  oTable.Range.Fields.Update

End Sub
(In der letzten Tabellenzeile und war in der ersten Spalte wird. die Formal "Sum Über" eingefügt.)



Anmerkung

· Setzen Sie für die Konstante mit dem Namen Ausdruck bei Word 97 den Wert Summe(Über) statt Sum(Über).

Hier noch eine elegantere (implizite) Lösung dieses Problems. Damit diese Mimik funktioniert, müssen in der Spalte allerdings schon Zahlen eingetragen sein. Diese Mimik taugt ohne Anpassung für alle MS Word-Versionen ab Word 97.



Sub FormelInLetzeTabellenSpalte2()

  Dim oTable As Table, Zelle As Cell

  Set oTable = ActiveDocument.Tables(1)

  oTable.Cell(oTable.Rows.Count, 1).AutoSum

  oTable.Range.Fields.Update

End Sub
(In der letzten Tabellenzeile und war in der ersten Spalte wird. die Formal «Sum Über» eingefügt.)



Verschachtelte Tabellen ab MS Word2000 verarbeiten

· Zurück zum Anfang
Ab Word2000 kann man nun, wie in HTML auch, verschachtelte Tabellen erstellen. Eine Zelle kann nun also statt Text oder Grafik eine andere Tabelle beinhalten. Dies lässt sich dann für viele Ebenen weiterziehen.

Im Folgenden finden Sie eine allgemeingültige Routine, um Zellen von verschachtelten Tabellen sequentiell abzuarbeiten. Dabei werden zuerst die Zellen der innersten Tabelle abgearbeitet. Dieses Vorgehen wird dann weiter gezogen bist zur äussersten Tabelle. Zur Sichtbarmachung des Resultates, wird Text in die Tabellenzellen eingefügt.

Erstellen Sie eine (mehrfach) verschachtelte Tabelle, Kopieren Sie den Code in ein neues Modul und testen Sie das Makro.



Private Zähler As Long
Sub Test()

  Zähler = 0

  RekursiveTabelleVerarbeiten ActiveDocument.Tables(1)

End Sub
Private Sub RekursiveTabelleVerarbeiten(oTable As Table)

  Dim oCell As Cell

  For Each oCell In oTable.Range.Cells

    If Not oCell.Tables.Count = 0 Then
      RekursiveTabelleVerarbeiten oCell.Tables(1)

    Else
      Zähler = Zähler + 1

      oCell.Range.Text = Zähler

    End If
  Next
End Sub
(Die Zellen einer verschachtelten Tabelle ab Word2000 werden sequentiell verarbeitet, wobei die innere Tabelle Priorität hat.)



· Bei dem eben aufgezeigten Codebeispiel handelt es sich um ein anschauliches Elaborat für eine Funktion, welche rekursiv aufgerufen wird. Bei einer rekursiven Funktion, wird die bereits aktive Funktion als Unterfunktion so lange immer wieder aufgerufen, bis der ganze Job getan ist.

Word-Tabelle auf einer UserForm abbilden

· Zurück zum Anfang
Folgendes Beispiel bildet die erste Tabelle im Hauptteil des aktiven Word-Dokuments auf einer UserForm ab.

Implementierung

Um die Lösung zu testen, verfahren Sie wie folgt: 

1. Erstellen Sie ein neues Word-Dokument und fügen sie eine Tabelle ein. Grundsätzlich kann diese Tabelle eine beliebige Anzahl von Zeilen und Spalten aufweisen. Das Programm ermittelt die Grösse der Tabelle dynamisch. Da die UserForm aber nicht beliebig gross werden sollte, ist die Grösse im Code bewusst limitiert (12 Zeilen bzw. 6 Spalten). Fühlen Sie sich aber frei, diese beiden Konstanten im Kopf des Prozesses mit dem Namen DynamischeSchaltflächenTabelleAufruf nach Ihren Bedürfnissen nach oben anzupassen. Die Namen der Konstanten lauten MaxZeilen bzw. MaxSpalten. Bedenken Sie aber dabei, dass die Anzahl der Controls, welche eine UserForm beherbergen kann nach oben limitiert ist.

2. Füllen Sie die Zellen der Tabelle mit etwas Text, damit das Ergebnis später besser sichtbar wird.

3. Wechseln Sie in den VBA-Editor und fügen Sie ein neues Module ein.

4. Kopieren Sie den Prozess mit dem Namen DynamischeSchaltflächenTabelleAufruf in dieses Modul. Beachten Sie, dass diese Routine davon ausgeht, dass die UserForm den Namen UserForm1 trägt. Allenfalls muss dieser Name von Ihnen noch angepasst werden.

5. Fügen Sie eine neue UserForm ein. Diese UserForm bleibt leer, da die Textfelder dynamisch, also mit Code erstellt werden.

6. Führen Sie einen Rechtsklick auf die UserForm aus und wählen Sie im Kontextmenü den Punkt Code anzeigen.

7. Löschen Sie die zwei Code-Zeilen, welche MS Word automatisch generiert hat und kopieren Sie den längeren Code, welcher für die UserForm bestimmt ist, in das nun leere Codefenster der UserForm.

8. Sie sind nun schon bereit, um die Sache zu testen.

VBA-Code

Aufruf der UserForm

Folgender Code tätigt die nötigen Vorarbeiten und ruft dann die UserForm auf. Kopieren Sie diesen Code in ein neues Modul.



Private Const MaxZeilen = 12
Private Const MaxSpalten = 6
Sub DynamischeSchaltflächenTabelleAufruf()

  If ActiveDocument.Tables.Count = 0 Then
    MsgBox "Im Dokument befindet sich keine Tabelle."

    End
  End If
  Dim Tabelle As Table

  Set Tabelle = ActiveDocument.Tables(1)

  AnzZeilen = Tabelle.Rows.Count

  If AnzZeilen > MaxZeilen Then AnzZeilen = MaxZeilen

  AnzSpalten = Tabelle.Columns.Count

  If AnzSpalten > MaxSpalten Then AnzSpalten = MaxSpalten

  ReDim x(AnzZeilen - 1, AnzSpalten - 1) As String
  For i = 0 To AnzZeilen - 1

    For j = 0 To AnzSpalten - 1

      tmp = Tabelle.Cell(i + 1, j + 1).Range.Text

      x(i, j) = Left(tmp, Len(tmp) - 2)

    Next j

  Next i

  UserForm1.DynamischeSchaltflächenTabelleUserForm x()

End Sub
(Die Tabelle wird in ein zweidimensionales Datenfeld geladen, anschliessend wird die UserForm aufgerufen.)



Der Code für die UserForm

Folgender Code erstellt dynamisch die nötigen Textfelder, um die Tabellenzellen abzubilden. Kopieren Sie die Zeilen in das Codefenster einer leeren UserForm mit dem Namen UserForm1.



Private Lab As MSForms.Label

Private TB As MSForms.TextBox

Sub DynamischeSchaltflächenTabelleUserForm(x() As String)

  Dim TB As MSForms.TextBox

  Titel = 24

  Header = 12

  Rand = 6

  StartOben = 3

  StartLinks = 6

  Höhe = 15
  Breite = 32
  AbsX = 9
  AbsY = 6
  Me.Height = ((UBound(x, 1) + 1) * Höhe) + (UBound(x, 1) * AbsY) + _

     (2 * StartOben) + Titel + Header

  Me.Width = ((UBound(x, 2) + 1) * Breite) + (UBound(x, 2) * AbsX) + _

     (2 * (StartLinks + Rand))

  oben = StartOben

  links = StartLinks + Rand

  For i = 0 To UBound(x, 2)

    Set Lab = Me.Controls.Add("Forms.Label.1", , True)

    With Lab

      .Top = oben

      .Left = links

      .Height = 9

      .Width = Breite

      .Caption = Chr(i + 65)

      links = links + Breite + AbsX

    End With
  Next i

  oben = StartOben + Header

  For i = 0 To UBound(x, 1)

  Set Lab = Me.Controls.Add("Forms.Label.1", , True)

  With Lab

    .Top = oben

    .Left = StartLinks

    .Height = 9

    .Width = 12

    .Caption = i + 1

    .TextAlign = fmTextAlignLeft

    oben = oben + Höhe + AbsY

  End With
  Next i

  oben = StartOben + Header

  For i = 0 To UBound(x, 1)

    links = StartLinks + Rand

    For j = 0 To UBound(x, 2)

      Set TB = Me.Controls.Add("Forms.TextBox.1", "TB" & Chr(j + 64) & i + 1, True)

      With TB

        .Top = oben

        .Left = links

        .Height = Höhe

        .Width = Breite

        .Text = x(i, j)

        links = links + Breite + AbsX

      End With
    Next j

    oben = oben + Höhe + AbsY

  Next i

  Me.Show

End Sub
(Bildet eine Word-Tabelle in einer UserForm ab. Die dazu nötigen Textfelder werden dynamisch erstellt. Grösse der Textfelder und die Abstände zwischen den Feldern können Sie im Kopf des Codes allenfalls noch anpassen.)



Weitere VBA-Splitter im Zusammenhang mit MS Word Tabellen

· Zurück zum Anfang
Folgende Links - zeigen aus sehr kompetenter Feder - weitere VBA-Splitter rund um Word-Tabellen auf:

Microsoft Word MVP FAQ Site

· Verarbeitungsgeschwindigkeit von Tabellen optimieren
· Doppelte Zeilen in einer Tabelle löschen
· Leere Zeilen in einer Tabelle löschen
· Feststellen, ob eine Tabellenzelle leer ist
· Alle Tabellenzeilen löschen, welche eine vorgegebene Zeichenfolge enthalten
· Index der Tabelle feststellen, in welcher sich die Einfügemarke befindet
Ein MS Excel-Tabellenblatt in einer Word-Tabelle abbilden

· Zurück zum Anfang
Im Folgenden finden Sie eine Routine, welche eine Excel-Tabelle in eine Word-Tabelle umwandelt (als Kopie). Die Tabelle wird am Ort der Einfügemarke in das Word-Dokument eingefügt.

Dank dem Einsatz von DAO (Data Access Object) kann diese Aufgabe sehr effizient gelöst werden, die Excel-Mappe muss dabei nämlich nicht mit einer trägen Office-Anwendung geöffnet werden.

Die Kennzeichnung, welche Daten übertragen werden sollen geschieht implizit, in dem Sie einen Benannten Bereich in der Excel-Tabelle definieren. Im Folgenden wird Schritt für Schritt erklärt, wie Sie konkret vorgehen sollten:

Implementierung

1. Öffnen Sie die MS Excel-Mappe mit den Daten, welche in einer Word-Tabelle abgebildet werden sollen.

2. Markieren Sie auf den Tabellenblatt den Bereich der Tabelle, welcher abgebildet werden soll.

3. Vom Hauptmenü in MS Excel wählen Sie Einfügen - Name - Definieren….
4. Vergeben Sie nun dem Bereich einen aussagekräftigen Namen. Speichern Sie dann die Excel-Mappe. Sie können die Excel-Anwendung jetzt schliessen.

5. Wechseln sie mit der Tastenkombination [Alt + F11] in den VBA-Editor.

6. Kopieren Sie den nachfolgenden Code in ein neues Modul.

7. Passen Sie nun im Kopf des Codes die Konstanten nach Ihren Gegebenheiten an. Die Konstante Pfad soll den Pfadname Ihrer Excel-Arbeitsmappe enthalten; und die Konstante Bereich soll den Bereichsnamen tragen, welchen Sie dem fraglichen Bereich in Excel vergeben haben.

8. Immer noch im VBA-Editor klicken Sie nun im Hauptmenü auf Extras - Verweise….
9. Suchen Sie in der Liste nach einem Eintrag Microsoft DAO und aktivieren Sie diesen Verweis. Falls Sie mehrere Einträge bezüglich DAO vorfinden, aktivieren Sie die höchste (sprich aktuellste) Version. Das folgende Beispiel wurde mit der Microsoft DAO 3.51 und Object Library Microsoft DAO 3.6 Object Library getestet.

10. Speichern Sie nun Ihr Werk und rufen Sie das Makro auf.

VBA-Code



Private Const Pfad = "C:\Eigene Dateien\Kantone der Schweiz.xls"

Private Const Bereich = "Kanton"

Sub ExcelTabelleInWordTabelle()

  Dim db As DAO.Database

  Dim tblObj As DAO.TableDef

  Dim rs As DAO.Recordset

  Set db = OpenDatabase(Pfad, False, True, "Excel 8.0;")

  DoEvents

  Q = Chr(34)

  Set rs = db.OpenRecordset("Select * FROM `" & Bereich & "`")

  For i = 0 To rs.Fields.Count - 1

    strT = strT & rs.Fields(i).SourceField & Chr(1)

  Next
  While Not rs.EOF

    strT = Left(strT, Len(strT) - 1)

    strT = strT & Chr(13)

    For j = 0 To rs.Fields.Count - 1

      strT = strT & rs.Fields(j).Value & Chr(1)

    Next j

    rs.MoveNext

  Wend
  strT = Left(strT, Len(strT) - 1)

  Selection.InsertBefore strT

  Selection.ConvertToTable Separator:=Chr(1), _

    Format:=wdTableFormatColorful2, AutoFit:=True
  Selection.Collapse Direction:=wdCollapseStart

  Set rs = Nothing
  Set tblObj = Nothing
  Set db = Nothing
End Sub
(Aus einem Benannten Bereich in einer Excel-Tabelle wird unter Einsatz von DAO eine Word-Tabelle. Die Excel-Arbeitsmappe braucht dabei nicht mit einer trägen Office-Anwendung geöffnet werden.)



Eine MS Access-Tabelle in einer Word-Tabelle abbilden

· Zurück zum Anfang
Einleitung

Im Folgenden finden Sie eine Routine, welche eine Access-Tabelle in eine Word-Tabelle umwandelt (als Kopie). Die Tabelle wird am Ort der Einfügemarke in das Word-Dokument eingefügt.

Dank dem Einsatz von ADO (ActiveX Data Object) kann diese Aufgabe sehr effizient gelöst werden, die MS Access MDB muss dabei nämlich nicht mit einer trägen Office-Anwendung geöffnet werden.

Das Beispiel bezieht sich auf eine MS Access Datenbank, welche Sie als Beispiel auf Ihrer MS Office CD vorfinden. Die Datei hat den Namen Nordwind.mdb. Ab der Office2000-Ausgabe kann diese mdb auch unter einem etwas anderen Namen laufen (Northwind.mdb).

Diese Datenbank besteht aus diversen Tabellen und Formularen. Für unser Beispiel habe ich die Tabelle mit dem Namen Bestellungen ausgewählt. Diese Tabelle wiederum beinhaltet ein kleines überschaubares Set an Datenfelder und etwa 800 Datensätze.

Es kommt eine SQL-Anweisung zur Anwendung, um festzulegen, welche Felder übertragen werden sollen. SQL ist das Akronym für Structured Query Language. Es handelt sich dabei glücklicherweise um eine herstellerunabhängige Methode, welche weite Verbreitung findet. Sie werden darüber - z.B. im WEB - sehr viel Literatur finden, um die Sache noch zu verfeinern. Auch im Skript Eine Prise SQL finden Sie zahlreiche Beispiele für die Syntax von SQL. Sie können mit einer einzigen SQL-Anweisung eigentlich alles Notwendige mitgeben, namentlich:

· Den Namen der Tabelle, welche Sie ansprechen möchten.

· Die Spalten (Felder) dieser Tabelle, welche Sie zurück haben möchten.

· Einfache oder auch anspruchsvolle Abfrageoptionen.

· Letztlich können Sie auch noch gleich die Sortierreihenfolge der Datensätze vorgeben.

Ergebnis

Das Ergebnis des Codes ist eine anschauliche und formatierte Tabelle, welche alle Datensätze in der Tabelle Bestellungen anzeigt. Aufgelistet werden folgende Feldinhalte: Lieferdatum, Empfänger, Ort und Bestimmungsland. Sie können diese Auswahl selbstverständlich Ihren Bedürfnissen anpassen.

Implementierung

Um das Beispiel einzurichten, gehen Sie bitte wie folgt vor:

1. Beschaffen Sie sich die Access-Datenbank Nordwind, welche Sie als Beispiel auf Ihrer Festplatte oder auf Ihrer MS Office-CD vorfinden.

2. Wechseln Sie mit der Tastenkombination [Alt + F11] in die VBA-Umgebung.

3. Dort angekommen, wählen Sie vom Hauptmenü den Dialog Extras - Verweise….
4. In der erscheinenden Liste aktivieren Sie den Verweis auf die Objektbibliothek mit dem Namen Microsoft ActiveX Data Objects x.x Object Library. Sollte die Liste mehrere Versionen von ADO anzeigen, wählen Sie die neuere, sprich höhere Version. Das folgende Beispiel wurde mit der Microsoft ActiveX Data Object 2.5 Library und Microsoft ActiveX Data Object 2.6 Library getestet.

5. Kopieren Sie den folgenden VBA-Code in ein neues Modul. Passen Sie den Pfadnamen der mdb im Kopf des Codes an und testen Sie das Makro.

VBA-Code



Private Const Pfad = "C:\Eigene Dateien\Nordwind.mdb"

Sub WordTabelleAusAccessDatenGenerieren()

  Dim oWord As Word.Application

  Dim oDoc As Word.Document

  Dim oRange As Word.Range

  Dim oConn As ADODB.Connection

  Dim oRS As ADODB.Recordset

  Set oRange = Selection.Range

  Set oConn = CreateObject("ADODB.Connection")

  oConn.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & _

     Pfad & ";Persist Security Info=False"

  Set oRS = oConn.Execute( _

    "SELECT Lieferdatum, Empfänger, Ort, Bestimmungsland FROM Bestellungen")

  sTemp = oRS.GetString(adClipString, -1, vbTab)

  sTemp = "Lieferdatum" & vbTab & "Empfänger" & vbTab & _

     "Ort" & vbTab & "Bestimmungsland" & vbCrLf & sTemp 'Tabellenkopf
  oRange.Text = sTemp

  oRange.ConvertToTable Separator:=vbTab, Format:=wdTableFormatColorful2

  Set oWord = Nothing
  Set oDoc = Nothing
  Set oConn = Nothing
  Set oRS = Nothing
End Sub
(Aus einer MS Access-Tabelle wird eine anschauliche und formatierte Word-Tabelle. Die Tabelle wird am Ort der Einfügemarke in das Word-Dokument eingefügt.)



Links zu diesem Thema

· Zurück zum Anfang
Eigene Links

· Wie kann ich in einem VBA-Programm (Makro) feststellen, ob sich die Einfügemarke in einer Tabelle befindet?
· VBA-technischer Umgang mit Textmarken
· Tabelle der Dokumentteile (StoryTypes) in MS Word
· VBA-technischer Umgang mit "Shapes" und "InlineShapes" 
· Mittels DAO Daten aus einem MS Excel-Tabellenblatt ziehen, ohne die Arbeitsmappe explizit zu öffnen
· Mittels ADO auf MS Access zugreifen
Microsoft Word MVP FAQ Site

· Maximising the performance of Word tables
· Deleting duplicate rows in a table
· Deleting all empty rows in a table using a macro
· Detect whether a table cell is empty
· Delete all rows of a table that contain a particular text string in the first column
· Determine the index number of the current paragraph, table, section ...
URL

http://mypage.bluewin.ch/reprobst/WordFAQ/Tabellen.htm

René Probst, April 2005

Änderungen

· 15. Juni 2000 - Erstellt.

· 19. Januar 2001 - Zeilen und Spalten hinzufügen

· 20. Januar 2001 - Arbeiten mit dem Range-Objekt von Tabellen

· 25. Februar 2001 - Tabellenspalten verarbeiten

· 19. Februar 2003 - Generell überarbeitet.

